

<http://www.youtube.com/watch?v=aboZctrHfK8>

42 Programming Language Meta-programming as Default

Motivation

Libraries:

- collections of data and behavior
- have well-defined interfaces
- suitable to be used by multiple independent programs.

Library usage dramatically increase programmers' productivity.

Today a well-designed program can use up to 10 third-party libraries. However, when more than 10 libraries are used keeping trace of the issues arising from those libraries and their interplay become very difficult.

Today, a goal for tomorrow

Today a programmer can use a library containing **millions of instructions** without the need to understand each instruction individually, and instead can rely on the library interface.

With 42, a programmer will be able to use **millions of libraries** without the need to understand each library individually, and may rely on libraries managing and organizing other libraries.

This would alter the perception of what programming is: Normally the programming language is considered “the thing the programmers use to code”. The philosophy of 42 is different: the language provides the foundation for libraries, and libraries is what programmers use to code.

Smaller libraries

Today libraries tends to be massive chunks of code with tons of functionalities packed into.

This is because importing and managing libraries is hard, so is convenient for libraries to "expand" their domain in related areas.

With easy - transparent - libraries, There will be no reason to have so many functionalities packed all together.

Decorators

Library composition is obtained using decorators, that are build upon basic decorators **Compose** and **Refactor::***.

Composing classes “as modules”,
with a associative and commutative compose/merge/sum operation:

- nested classes with same name are recursively merged;
- the set of implemented interfaces is merged;
- methods with same name are merged, and conflict arises if both version have an implementation or if the headers are not compatible.

Good, but... how to merge fields/constructors?

Constructors and fields

What is exactly a field/constructor in 42,
since I can only access them as getters/setters/factories?
What is the exact meaning of writing `Point: { (N x, N y) }`?
It is just sugar for

```
Point: {   '(N x, N y) desugar into
  type method Outer0 (N x, N y)
  method N x ()
  method N y ()
}
```

That is, a class with just abstract methods.

```
Point:{  '(N x, N y) desugar into
  type method Outer0 (N x, N y)
  method N x()
  method N y()
}
```

In many languages, a concrete class have no abstract methods; In 42, a concrete class have a **coherent set** of abstract methods:

- A class with no abstract methods is concrete (as `java.lang.Math`)
- A class with a single abstract **type** method returning `Outer0` whose parameters are all of placeholder type (as `N^ x`) is concrete.
- If such abstract **type** method is present, then, for each parameter, it is allowed to have abstract getters and setters, where the name maps back to the parameter name.

Take a deep breath, to me this is game changing and mind blowing. There are only methods. The constructor/getter implementation is not protected/hidden/secret/native. It is just **not there**. The whole formal model works without requiring any difference between abstract methods and getters/setters/constructors.

Composing Constructors and fields

```
C:{
  type method Library point() {(N x, N y)
    method N distance(Outer0 that){
      x=this.x()-that.x()
      y=this.y()-that.y()
      return (x*x+y*y).sqrt()
    }
  }
}

ColPoint:Compose[C.point()]<<{(N x, N y, Color color)
  type method Outer0 (N x,N y)
    Outer0(x:x,y:y,color:Color.black())
}
```


Composing Constructors and fields

```
C:{  
  type method Library point() {(N x, N y)  
    method N distance(Outer0 that){  
      x=this.x()-that.x()  
      y=this.y()-that.y()  
      return (x*x+y*y).sqrt()  
    }  
    method Outer0 clone(N x) Outer0(x:x,y:this.y())  
  }  
}  
ColPoint:Compose[C.point()]<<{(N x, N y, Color color)  
  type method Outer0 (N x,N y)  
    Outer0(x:x,y:y,color:Color.black())  
  } 'now with sets the color to black :-('
```

Composing Constructors and fields

```
C:{
  type method Library point() {(N x, N y)
    method N distance(Outer0 that){
      x=this.x()-that.x()
      y=this.y()-that.y()
      return (x*x+y*y).sqrt()
    }
    method Outer0 clone(N x)
  }
}
ColPoint:Data[]<<Compose[C.point()]<<{(N x, N y, Color color)
  type method Outer0 (N x,N y)
    Outer0(x:x,y:y,color:Color.black())
} 'let data do its job
```

Traits, trivially

Traits are just groups of methods
(and they can implement interfaces)

```
{<:Foo,Bar ..method T x().. method Void foo().. }
```

Just add traits together to get other threads.

```
Compose[Lib.a();Lib.b();..]
```

Sum a trait to a (concrete) class and get as a result a better (concrete) class, if such class offers all the methods required (i.e. abstract) in the thread.

```
Compose[Lib.a();Lib.b();..]<<{..}
```

Expression problem as matrix

	toString	eval
Num	<pre>{ Exp:{interface<:HasToString } Num:{(N that)<: Exp method toString() S""++this.that() } }</pre>	<pre>{ Exp:{interface<:HasEval } Num:{(N that)<: Exp method eval() this.that() } }</pre>
Sum	<pre>{ Exp:{interface<:HasToString } Sum:{(Exp left, Expr right)<: Exp method toString() this.left().toString() ++this.right().toString() } }</pre>	<pre>{ Exp:{interface<:HasEval } Sum:{(Exp left, Expr right)<: Exp method eval() this.left().eval() ++this.right().eval() } }</pre>
UMinus	<pre>{ Exp:{interface<:HasToString } UMinus:{(Exp that)<: Exp method toString() S"-"+this.that().toString() } }</pre>	<pre>{ Exp:{interface<:HasEval } UMinus:{(Exp that)<: Exp method eval() this.that().eval()*-1N } }</pre>

Expression problem as matrix

Just sum the desired line/rows together

```
MySolution:ExprProblem(  
  useRows:SList[S"uMinus";S"sum"]  
  useColumns:SList[S"eval";S"toString"]  
)
```

Refactor

Ok, **Compose** is cool. What about **Refactor::***? It can rename methods, nested classes, and can set members private. More than that, **Refactor::*** is cool since it can apply **Compose** internally.

```
D:Refactor::RenamePath[Path"C" to: Path"K"]<<
  RenamePath[Path"A" to: Path"K"]<<{
    C:{method N c() }
    A:{method N a() }
    method C m(A a)
  }
```

rewrites to

```
D:{
  K:{method N c() method N a() }
  method K m(K a)
}
```

Refactor

```
D:RenamePath[Path"C::A" to: Path"Outer0" ] << {  
  method N top()  
  C:{method N c()  
    A:{ method N a() }  
  }  
}
```

rewrites to

```
D:{  
  method N top()  
  method N a()  
  C:{method N c() }  
}
```

```
D:RenamePath[Path"Outer0" to: Path"C::A" ] << {..}
```

rewrites to

```
D:{ C:{ A:{..}}}
```

Redirect

```
D:Refactor::Redirect[Path"C" to:S ] << {  
  C:{... B:{...}}  
  method C m(C::B x)  
}
```

rewrites to

```
D:{  
  method S m(S::B x)  
}
```

This last operation requires to check the structural type of s and $s :: B$ to be compatible with c and $c :: B$.

Towel Intro

A towel is about the most massively useful thing a programmer can have. A towel has immense psychological value, and you should always know where your towel is.

```
{reuse L42.is/BasicTowel
Main:{
  Debug(S"Hello world")
  return ExitCode.normal()
}
```

Here `Debug`, `S` and `ExitCode` are classes defined inside the towel `BasicTowel`. You can code without a towel, but this means starting from first principles, which could be quite unpleasant; especially since the only primitive things that 42 offers are Library literals (code as first class entities), the constant `void`, and the types `Library`, `Void` and `Any`.

Towel Intro

We generally call towels libraries that are expected to provide standard functionalities and types, such as numbers `N`, booleans `Bool`, strings `S` and various kinds of system errors (and `Load`, as discussed later).

However, we do not expect all L42 programs to reuse the same towel. For hygienic reasons, in real life everyone tends to use their own towel. For similar reasons, any 42 program will use its own towel.

We expect different programs to use massively different libraries for what in other languages is the standard library. That is, there is no such thing as "the L42 standard library"

Towel Load

Most L42 libraries are not towels. All 42 libraries are closed code. Thus, all non-towel 42 library are generic (have abstract classes/methods)

```
{reuse L42.is/BasicTowel
Gui:Load[]<<{reuse L42.is/2dGui}
Main:{
  Gui.alert(S"Hello world")
  return ExitCode.normal()
}
```

`Load[]<<{..}` would intuitively behave like

`Refactor::Redirect[Path"N"to:N]<<.. "S"to:S..<<{..}`

`Load` redirect such abstract classes to their incarnation in the Towel, assuming they respect some structural interface.

Towel Multiple Towels

Towels shines when multiple towels are used at the same time.

```
{reuse L42.is/AdamsTowel
  ' here you can access to lots of
  ' utility classes declared inside the towel
  ' including numbers, strings and so on.
C:{reuse L42.is/FordTowel
  'here you can access a different set of classes.
  'For example, N would refer to the number in FordTowel
  'and to see the number declared in AdamTowel
  'you have to write Outer::N
}
}
```

Different code parts reason about different set of classes; including those predefined in other languages.

Useful for code that reasons on code; that is a very common task in L42.

Towel Multiple Towels

Define and deploy our own towel: like BasicTowel, with Gui as before.

```
{reuse L42.is/DeploymentTowel
Main:DeployTowel[URL".../GuiTowel";permissions:S"..."]<<{
  reuse L42.is/BasicTowel
  Gui:Load[]<<{reuse L42.is/2dGui}
}
}
```

DeploymentTowel is just some variation of BasicTowel, with Deployment functionalities already included. Here `Load` comes from BasicTowel, and the nested code literal results in closed code, thus `Deploy` will be able to correctly export it and save it in the provided URL, if we have permission.

Towel Embroidery

We can mix and match different implementations for numbers and strings (as an example). This requires some special care:

- strings should know their length (as a number)
- numbers should be comparable (into true/false booleans)
- booleans (and numbers) should be representable (as strings)

```
{reuse L42.is/DeploymentTowel
Main:(
  var code={ 'code with alternative versions for N and S
    reuse L42.is/BasicTowel
    NAlt:Load[]<<{reuse L42.is/InfinitePrecisionNatural}
    SAlt:Load[]<<{reuse L42.is/UTF8Strings} }
  code:=Refactor::RemoveImplementation[Path"N"]<<code
  code:=Refactor::RemoveImplementation[Path"S"]<<code
  'now N and S are abstract
  code:=Refactor::Rename[Path"NAlt" to:Path"N"]<<code
  code:=Refactor::Rename[Path"SAlt" to:Path"S"]<<code
  DeployTowel[URL".../MyTowel";permissions:S"..."]<<code
)
```

Library Deployment

Non-towel libraries can also be deployed in a similar way:

```
{reuse L42.is/DeploymentTowel
Main: (
  var code={
    reuse L42.is/BasicTowel
    MyLib:... 'this refers to the file MyLib.L42
  }
  code:=Refactor::ExposeAsLibrary[Path"MyLib"] <<code
  code:=Optimize::RemoveUnreachableCode[] <<code
  Deploy[URL"..MyTowel";permissions:S".."] <<code
)
```

Where `ExposeAsLibrary[Path"MyLib"] <<code` put the path `MyLib` at top level and make abstract classes like `S`, `N` and `Bool`. All the rest of the code is marked as private.

Then, `Optimize::RemoveUnreachableCode[] <<code` remove all the unreachable private code.

ExposeAsLibrary example code

This code show how error managing can be mangle with the behavioural logic in a bearable way.

```
method 'here this.target() is the path of the lib we want to expose
Library <<(Library that)exception TargetUnavailable,PrivacyCoupuled,InvalidShape{
  this.checkValid(that)'throws TargetUnavailable,InvalidShape

  var res=that
  res:=RemoveImplementationPath[Path"N"]<<res
  res:=RemoveImplementationPath[Path"S"]<<res
  res:=RemoveImplementationPath[Path"Bool"]<<res'leak PrivacyCoupuled
  catch exception (on Report::TargetUnavailable error Report::CodeNotReachable())

  res:=RenamePath[Path"N" to:this.target()++Path"N"]<<res
  res:=RenamePath[Path"S" to:this.target()++Path"S"]<<res
  res:=RenamePath[Path"Bool" to:this.target()++Path"Bool"]<<res

  res:=RenamePath[Path"Outer0" to:Path"PrImpl"]<<res
  res:=RenamePath[Path"PrImpl"++this.that() to:Path"Outer0"]<<res
  res:=MakePrivatePath[Path"PrImpl"]<<res
  return res
  catch exception (on Report error Report::CodeNotReachable())
}
```


checkValid

```
method Void checkValid exception TargetUnavailable, InvalidShape (  
  intro=Introspection(lib:that)  
  if !intro.hasPath(this.target()) (exception TargetUnavailable())  
  if !intro.hasPath(Path"N") (exception InvalidShape"..")  
  if !intro.hasPath(Path"S") (exception InvalidShape"..")  
  if !intro.hasPath(Path"B") (exception InvalidShape"..")  
  if intro.hasPath(this.target()++Path"N") (exception InvalidShape"..")  
  if intro.hasPath(this.target()++Path"S") (exception InvalidShape"..")  
  if intro.hasPath(this.target()++Path"B") (exception InvalidShape"..")  
  void)
```

Support for decent size applications

```
{reuse L42.is/AnotherDeploymentTowel
Local:UrlData.readOnly(prefix:URL"file:///./myProjects/")
Remote:UrlData.writeOnly(prefix:URL"L42.is/byMarco/",permission:S"..")
Web:UrlData.writeOnly(prefix:URL"www.myWebsite.com/",permission:S"..")
Header:Deployer[]<<{ ()
  method DeployResult() {
    var acc=Project(Local"header")
    b= Benchmark(Local"headerBenchmark")
    acc :=b.optimize(acc, inlining:10Minute)
    acc :=b.optimize(acc, parallel:1Minute)
    return acc.deploy(Remote"header")
  }
}
GraphicEngine:Deployer[]<<{ (Header h)
  method DeployResult() {
    var acc=Project(Local"graphicEngine")
    'inside reuse L42.is/byMarco/header
    b1=Benchmark(Local"graphicEngineBenchmark1")
    b2=Benchmark(Local"graphicEngineBenchmark2")
    acc =b1.optimize(acc, inlining:10Minute)
    acc =b2.optimize(acc, parallel:10Minute)
    return acc.deploy(Remote"graphicEngine")
  }
}
```

Support for decent size applications

```
MyGame:Deployer[]<<{ (Header h GraphicEngine e)
  method DeployResult() {
    var acc=Project(Local"myGame")
    b=Benchmark(Local"myGameBenchmark")
    acc =b.optimize(acc, inlining:10Minute)
    acc =b.optimize(acc, parallel:10Minute)
    return acc.deploylist() [
      windowsExe: acc Web"myGame.exe";
      linuxExe: acc  Web"myGame";
      javaExeJar: acc Web"myName.jar";
      htmlPage: Local"index.html".asText()
      name: Web"index.html";

    ]
  }
}

Main:Build(MyGame) 'Build compute dependency graph and check for changes
```

Getting started with Active Libraries

```
reuse L42.is/AdamsTowel
AgeInput: Gui.input(receive:N, request:S"insert age")
  'class AgeInput is the first class we declare!
  'is generated by method "Gui.input"
Main:{
  Gui::Alert(S"your age is "+AgeInput().asText())
  'class AgeInput principal method (the one with no name)
  'ask the user to insert a number N
  return ExitCode.normal()
}
```

Getting started with Active Libraries

```
reuse  L42.is/base
BirthInput: Gui.input (
  title: S"Date of birth"
  receive: Date   'we ask a widgets for the type Date
  request: S"insert your date of birth")
Main: {
  Date userDate= BirthInput ()
  Gui::Alert (S"your date of birth is "+userDate.asText ())
  return ExitCode.normal ()
}
```

`Gui.input` can generate widgets for each type.

Getting started with Active Libraries

```
reuse L42.is/base
Tables: DB(connection:DBConnection"....")
  'create a class with a nested for each db table
  'plus utility features to connect with the db
...
me=Tables::Student(id:....)
  'get student from id, queries also available.
....me.name()....
```

DB turns a db into classes. If the db is updated, and `name` is renamed in `fullName`, 42 will point to the error place and we will be guided into correspondingly updating our program.

```
reuse L42.is/base
Tables: DB(connection:....) ' with the updated db-schema
...
me=Tables::Student(id:....)
....me.name().... 'now I get a red sign here!
```

Can you guess what this code is doing?

```
{reuse  L42.is/AdamsTowel
Db: Load[]<<{L42.is/Db}
Gui: Load[]<<{L42.is/2dGui}
Tables: DB(S"...my connection string...")
StudentList: Collections.list(Tables::Student)
StudentsView: Gui.tableOf(StudentList)
QueryFrom: Tables::Query[result:StudentList](
  DB::SQL"Select * from Student where country=@country")
Main:{
  connection=Tables.connect()
  fromItaly=QueryFrom(connection, country:S"Italy")
  Gui.show(StudentsView(fromItaly))
  return ExitCode.normal()
}
```

Visualize DB

```
DB: Load[] << {L42.is/DB}
Gui: Load[] << {L42.is/2dGUI}
Tables: DB(S"...my connection string...")
StudentList: Collections.list(Tables::Student)
StudentsView: Gui.tableOf(StudentList)
QueryFrom: Tables::Query[result:StudentList](
  DB::SQL"Select * from Student where country=@country")
```

- `Load[] << {L42.is/..}` `Load` is loading the code of a library and, according to the conventions in *L42.is/AdamsTowel*, is linking numbers, strings and other general purposes classes to their correspondent inside of *L42.is/AdamsTowel*.
- `DB(S"...my connection string...")` creates classes to reify tables in the database.
- `StudentList: Collections.list(Tables::Student)` creates a collection for students.

Visualize DB

```
StudentsView: Gui.tableOf(StudentList)
QueryFrom: Tables::Query[result:StudentList](
  DB::SQL"Select * from Student where country=@country")
```

- **StudentsView: Gui.tableOf(StudentList)** create a widget visualizing lists of students. Every collection can be seen as a table, if the object supports iteration over its fields and its class supports iteration over its field names and types. **Data[]** can easily add those methods for you.
- **QueryFrom: Tables::Query[result:StudentList](..)** creates a (class that can only perform a) prepared query. If the query is of the form **"Select * from TableName where ..."** then it can produce a list of instances of a class under **Table::..**. Otherwise, it would also generate a nested class for the result.

Visualize DB

```
Main:{  
  connection=Tables.connect()  
  fromItaly=QueryFrom(connection, country:S"Italy")  
  Gui.show(StudentsView(fromItaly))  
  return ExitCode.normal()  
}
```

- **Tables.connect()** connects to the database.
- **fromItaly=QueryFrom(connection, country:S"Italy")**
Using **QueryForm** we ask for students coming from Italy. Notice how **country:** is part of the method signature. The name of the parameter was extracted while generating the class. A good IDE autocompletion would take advantage of it.
- **Gui.show(StudentsView(fromItaly))** close the circle, showing a table view of our Italian students.

Maintainability?

```
{reuse  L42.is/AdamsTowel
Db: Load[]<<{L42.is/Db}
Gui: Load[]<<{L42.is/2dGui}
Tables: DB(S"...my connection string...")
StudentList: Collections.list(Tables::Student) ' HERE!!
StudentsView: Gui.tableOf(StudentList)
QueryFrom: Tables::Query[result:StudentList](
  DB::SQL"Select * from Student where country=@country")
Main:{
  connection=Tables.connect()
  fromItaly=QueryFrom(connection, country:S"Italy")
  Gui.show(StudentsView(fromItaly))
  return ExitCode.normal()
}
```